PharmaSUG 2014 - Paper HA06

The Association of Morbid Obesity with Mortality and Coronary Revascularization among Patients with Acute Myocardial Infarction

Ashwini S Erande MPH University Of California Irvine, Orange CA

Shaista Malik MD PhD University Of California Irvine, Orange CA

ABSTRACT

The aim of this study was to investigate the impact of morbid obesity (body Mass index >= 40 kg/m2) on in-hospital mortality and coronary revascularization outcomes in patients presenting with acute myocardial infarctions (AMI).The Nationwide Inpatient Sample of the Healthcare Cost and Utilization Project was used, and 413,673 Patients hospitalized with AMIs in 2009 were reviewed. Morbidly obese patients constituted 3.7% of all patients with AMIs. All the analyses were performed in SAS® 9.3.

ARRAY Statements were used to create the morbid obesity variable based on the ICD9 codes from 24 "Diagnosis" data elements. The SAS procedures PROC SURVEYFREQ and PROC SURVEYLOGISTIC were used to perform bivariate and multivariate analyses on this sample survey database.

The unadjusted and adjusted analyses performed using PROC SURVEYFREQ and PROC SURVEYLOGISTIC respectively, revealed that morbidly obese patients compared with those not morbidly obese were more likely to undergo any invasive coronary procedures when presenting with either ST-segment elevation myocardial infarction and also have a higher mortality rate.

The SAS procedures used to analyze and summarize the data within this context are presented in this paper.

INTRODUCTION

This study has explored the association of worse short-term outcomes for morbidly obese patients presenting with acute coronary syndromes in Nationwide Inpatient Sample (NIS) database of the Healthcare Cost and Utilization Project. The study analyzed the associations among morbid obesity, treatment utilization, and mortality while adjusting for baseline characteristics, including co-morbidities, for 413,673 patients hospitalized with acute myocardial infarctions. This study involved a population-based sample of all patients admitted with AMIs to 1,045 hospitals in 44 states in 2009 whose admission and discharge data were included in the NIS.

We used SAS version 9.3 (SAS Institute Inc., Cary, North Carolina) for all analyses. ARRAY statements were used to create the main diagnosis variable. Given the finite (survey) nature of the data, PROC SURVEYFREQ and PROC SURVEYLOGISTIC were used to perform bivariate and multivariate analyses to obtain the chi squares, p-values and Odds ratio with Confidence Intervals respectively.

PROC SURVEYFREQ with the chi square option was used to find differences between morbidly obese and normal (not morbidly obese) patients with respect to the categorical variables; coronary revascularization procedure utilization and mortality experienced post those procedures and to obtain distributional statistics.

PROC SURVEYLOGISTIC is similar to logistic and other regression procedures; however it is designed to handle the sample survey data and thus incorporates the sample design information into the analyses. It produced the estimates, adjusted odds ratios (ORs) for in-hospital mortality as well as procedure use that was performed. To control for differential characteristics of morbidly obese patients and those not morbidly obese, covariates including age, gender, race, income, Elixhauser co-morbidities, and hospital characteristics such as hospital location, hospital control (profit or nonprofit), hospital teaching status, and hospital volume were included in the models. All analyses were weighted using NIS-provided weights to create national estimates.

This paper discusses the above listed SAS procedures and walks through them as they help generate the results that demonstrate how the patients with morbid obesity had higher coronary revascularization procedure use and lower odds of in-hospital mortality, compared to those not morbidly obese, consistent with the phenomenon of the "obesity paradox."

STUDY SAMPLES AND DATA ELEMENTS

The key interest variable in the study was 'morbid obesity'. Other explanatory variables in the risk adjustment algorithm include the socio demographic variables, hospital factors and 30 chronic co-morbidities. These are specified in the multivariate analyses section.

The primary sample contained - AMI (Acute Myocardial Infarction) cases. Sub-Samples consisted of STEMI (ST Elevated MI) and NSTEMI (Non- ST Elevated MI) cases

Response Variables: The principal outcome measure was short-term all-cause mortality (in-hospital mortality), which was defined as death that occurred during the initial hospitalization, between the day of hospital admission and date before discharge. Secondary outcomes included utilization of coronary procedure and those were diagnostic coronary angiography, percutaneous coronary intervention (PCI) or coronary artery bypass graft (CABG) surgery.

These data include ICD-9-CM-coded primary and secondary diagnoses; primary and secondary procedures; admission and discharge status; demographic information such as gender, age, race and ethnicity, and median income for ZIP code divided into quartiles; expected payment source; total charges; length of stay; and hospital region, teaching status, ownership type, and bed size. We used ICD-9-CM secondary diagnosis codes and a database-defined variable for morbid obesity (body mass index _40 kg/m2) developed by the Agency for Healthcare Research and Quality. ICD-9-CM secondary codes were used to indicate the presence of up to 30 chronic co-morbidities likely to have been present on admission, using the Elixhauser comorbidity adjustment method developed at the Agency for Healthcare Research and Quality.

PROCEDURES FOR DATA TRANSFORMATION AND BIVARIATE ANALYSES.

In this section I have briefly introduced the Array and Do Loop statements to show their use in creating the main variable of interest that is Morbid obesity.

ARRAY AND DO LOOP STATEMENTS TO CREATE THE 'MORBIDOBESITY' VARIABLE:

Below is the SAS code that was used to include ARRAY Statement and Do Loop to create the Morbid Obesity variable, using 25 data elements that indicated primary, secondary and tertiary diagnoses.

```
data Obesity;
set Obesity;
array dxmorbidobe{24} dx2-dx25; ❶
morbidobesity=0;
do i=1 to 24; ❷
if dxmorbidobe{i} in ('27801') then morbidobesity=1;
end;
run;
```

The Array is defined as 'dxmorbidobe' and is assigned 24 members; dx2 to dx25 in step ❶ (which are the 'Diagnoses' variables in the dataset) to create the morbidobesity' variable . Dx1 is not used here as it is a primary diagnosis and is already used to define the main study sample variable called" Acute Myocardial Infarction.

In Step ❷, the 'if' statement indicates the condition when the variable 'morbidobesity' is coded 0 or 1 in step. And the Do loop (do i=1 to 24;) is employed to check if the condition is satisfied for each of the 24 diagnoses variables for every record before assigning 0 or 1 to the newly created morbidobesity' variable. Thus the newly created variable 'morbidobesity' reads as 1=morbid obesity present and 0=Absence of morbid obesity.

PROC SURVEYFREQ FOR BIVARIATE ANALYSES:

In bivariate analyses, PROC SURVEYFREQ: shows the association between 2 categorical variables by providing numbers and percentages.

PROC SURVEYFREQ: To compare co-morbidities in between morbidly obese patients and the non-morbidly obese patients.

```
Proc surveyfreq data= Obesity;
 tables   morbidobesity* (race1 income  diabetes hypertension renalfail perivasc
 hyperlipidemia) / row chisq; ❶
strata nis_stratum; cluster hospid ❷;weight discwt;
title 'Morbidobesity by race income and by main comorbidities'; run;
```

In Step ❶, the 'morbid obesity' variable compared with each of the variables listed in the parentheses separately. The use of option 'chisq' generates the chi square and p-value (highlighted in the output 1) that can be reported as baseline results.
The STRATA statement lists the variables that form the strata in a stratified sample design. The CLUSTER statement specifies cluster identification variables in a clustered sample design. ❷
The Weight statement uses the variable 'discwt' to provide weight to each observation in the input dataset to accomplish the weighted analyses.

Below in Output 1, frequencies, weighted frequencies and row percentages, along with the standard deviations and errors of frequencies and percentages respectively are displayed. In this case, the row percentages are of our interest. The Row Pct. gives the percent of observations in the row; for instance; percentage of cases of morbid obesity with diabetes will be 1931*100 /(1931+1108)= 63.44% ❸and ❹.

The SURVEYFREQ Procedure
Table of Morbidobesity by Diabetes

Data Summary	
Number of Strata	60
Number of Clusters	649
Number of Observations	82173
Sum of Weights	413673.094

Table of morbidobesity by dm_all_comorbid								
Morbidobesity	Diabetes	Frequency	Weighted Frequency	Std Dev of Wgt Freq	Percent	Std Err of Percent	Row Percent	Std Err of Row Percent
No	No	53086	267180	10928	64.5873	0.3853	67.0601	0.3719
	Yes	26048	131239	5357	31.7252	0.3533	32.9399❸	0.3719
	Total	79134	398419	15976	96.3125	0.1397	100.000	
Yes	No	1108	5577	380.96182	1.3482	0.0616	36.5614	0.9255
	Yes	1931	9677	618.59614	2.3393	0.0948	63.4386 ❹	0.9255
	Total	3039	15254	956.15063	3.6875	0.1397	100.000	
Total	No	54194	272757	11211	65.9355	0.3787		
	Yes	27979	140916	5857	34.0645	0.3787		
	Total	82173	413673	16746	100.000			

Rao-Scott Chi-Square Test	
Pearson Chi-Square	1208.6126
Design Correction	0.8276
Rao-Scott Chi-Square	1460.4436
DF	1
Pr > ChiSq	<.0001 ❺
F Value	1460.4436
Num DF	1
Den DF	589
Pr > F	<.0001
Sample Size = 82173	

Output 1. Portion Of The Output From PROC SURVEYFREQ With Chi Square

The Rao-Scott chi-square test is a design-adjusted version of the Pearson chi-square test, which involves differences between observed and expected frequencies. For two-way tables, the null hypothesis for this test is no association between the row and column variables.

Within the scope of this paper, we will consider only Rao-Scott and its p value; which are 1460.4 and <.0001 respectively ❺ . These values tell us that there is an association between these two variables and that significantly higher number of cases with morbid obesity have diabetes ❹as compared with those without morbid obesity ❸.

Following two examples are the codes and output of morbid obesity tested against two outcome variables of interest.

1. PROC SURVEYFREQ: To compare any coronary revascularization procedure utilization between morbidly obese patients and the non-morbidly obese patients.

```
proc surveyfreq data= Obesity;
tables morbidobesity*anyprocedure/ row chisq;
strata nis_stratum; cluster hospid ;
where stemi=1; ❻ weight discwt;
title 'Chisquare of morbidobesity vs Any procedure use in STEMI  patients'; run;
```

The SURVEYFREQ Procedure

Table of morbidobesity by Anyprocedure

Data Summary	
Number of Strata	57
Number of Clusters	486
Number of Observations	26581
Sum of Weights	134032.319

Table of morbidobesity by anyproc

morbidobesity	anyproc	Frequency	Weighted Frequency	Std Dev of Wgt Freq	Percent	Std Err of Percent	Row Percent	Std Err of Row Percent
No	No	1588	8042	430.95850	5.9998	0.3473	6.1664	0.3574
	Yes	24273	122368	5604	91.2971	0.3796	93.8336❼	0.3574
	Total	25861	130409	5721	97.2969	0.1390	100.000	
Yes	No	19	94.50659	21.57866	0.0705	0.0163	2.6085	0.5987
	Yes	701	3529	255.10712	2.6326	0.1375	97.3915 ❽	0.5987
	Total	720	3623	257.20010	2.7031	0.1390	100.000	
Total	No	1607	8136	434.17294	6.0703	0.3507		
	Yes	24974	125896	5777	93.9297	0.3507		
	Total	26581	134032	5896	100.000			

Rao-Scott Chi-Square Test	
Pearson Chi-Square	15.5216
Design Correction	0.9910
Rao-Scott Chi-Square	15.6633
DF	1
Pr > ChiSq	<.0001❾
F Value	15.6633
Num DF	1
Den DF	429
Pr > F	<.0001
Sample Size = 26581	

Output 2. The Difference Between Morbidly Obese And Non-Morbidly Obese Cases Presenting With STEMI In Getting Any Coronary revascularization Procedure

In step ❻ of the code, the ‘where’ statement is used to define the sub-set of the main input dataset; in this case; using ONLY the STEMI cases to find the association.

The SAS code above produces the result with Rao-Scott Chi-Square and p –value to show the difference in use of coronary catheterization between the morbidly obese and not-morbidly obese cases presenting with STEMI.

The Output 2 shows us the unadjusted results and using the row percentages can be interpreted as follows: a significantly higher number of morbidly obese patients ❽ were undergoing any invasive coronary procedures when presenting with STEMI (97.4% vs 93.8%, p <0.0001 ❾) compared with those not morbidly obese ❼.

2. PROC SURVEYFREQ: to compare in-hospital mortality in morbidly obese patients and the non-morbidly obese patients.

```
proc surveyfreq data= Obesity;
tables morbidobesity*Died/ row chisq;
strata nis_stratum; cluster hospid ;
weight discwt;
title 'Chisquare of morbidobesity vs Mortality use in AMI  patients'; run;
```

The SURVEYFREQ Procedure

Table of morbidobesity by DIED

Data Summary	
Number of Strata	60
Number of Clusters	649
Number of Observations	82173
Sum of Weights	413673.094

Table of morbidobesity by DIED

morbidobesity	DIED	Frequency	Weighted Frequency	Std Dev of Wgt Freq	Percent	Std Err of Percent	Row Percent	Std Err of Row Percent
No	No	74814	376633	15250	91.0461	0.1639	94.5320	0.1085
	Yes	4320	21786	838.29703	5.2664	0.1052	5.4680 ❿	0.1085
	Total	79134	398419	15976	96.3125	0.1397	100.000	
Yes	No	2934	14725	926.50779	3.5595	0.1354	96.5295	0.3496
	Yes	105	529.40356	61.74715	0.1280	0.0138	3.4705 ❿	0.3496
	Total	3039	15254	956.15063	3.6875	0.1397	100.000	
Total	No	77748	391358	15995	94.6056	0.1069		
	Yes	4425	22315	863.17565	5.3944	0.1069		
	Total	82173	413673	16746	100.000			

Rao-Scott Chi-Square Test	
Pearson Chi-Square	22.8176
Design Correction	1.0676
Rao-Scott Chi-Square	21.3731
DF	1
Pr > ChiSq	<.0001
F Value	21.3731
Num DF	1
Den DF	589
Pr > F	<.0001
Sample Size = 82173	

Output 3. Difference In In-Hospital Mortality Between Morbidly Obese And Non-Morbidly Obese Cases.

The code shown above produces the results with chi square and p –value to show the difference in mortality in morbidly obese and not-morbidly obese cases presenting with acute myocardial Infarction; the whole sample.

The Output 3 shows lower unadjusted in-hospital mortality rate in morbidly obese patients compared with those not morbidly obese in the overall acute myocardial infarction sample (5.5% vs 3.5%, p <0.0001) ➓.
After running PROC SURVEYFREQ for rest of the invasive coronary procedures against the 'morbid obesity', separately, we get the same output as shown in Output 2. Similarly, the PROC SURVEYFREQ for mortality post each coronary procedure against the 'morbid obesity' separately yields the results as shown in Output 3.
Using these outputs generated for each coronary procedure, the simplified result tables reporting the chi-square and p value are put together; shown below in Output 4.
These unadjusted results show significantly higher procedure utilization in morbidly obese patients compared to non-morbidly cases in section❶; while the section ❷ shows significantly lower in-hospital mortality post coronary revascularization procedures in morbidly obese cases as compared to their counterparts.

❶ Coronary Revasc. Procedure	Morbid obesity (AMI patients)					Morbid obesity (STEMI patients)				
	n	Yes	No	Chi - square	P - Value	n	Yes	No	Chi - square	P - Value
Any procedure	351,944	88.3 (13,475)	85.0 (338,468)	132.7	<0.0001	125,896	97.4 (3,529)	93.8 (122,368)	78.3	<0.0001
Diagnostic cath	88,250	24.7 (3,763)	21.2 (84,487)	105	<0.0001	12,123	10.3 (373)	9.01 (11,750)	7.1	0.008
PCI	217,492	45.1 (6,878)	52.9 (210,615)	356.1	<0.0001	102,266	72.7 (2634)	76.4 (99,632)	26.6	<0.0001
CABG	46,201	18.6 (2,834	10.9 (43,367)	877.2	<0.0001	11,508	14.4 (521)	8.4 (10,986)	159.8	<0.0001

	Morbid obesity (AMI patients)					Morbid obesity (STEMI patients)				
❷ Mortality post coronary procedure	n	Yes	No	Chi - square	P - Value	n	Yes	No	Chi - square	P - Value
Any procedure	10,884	12.1 (216)	17.8 (10,669)	38.3	<0.0001	2,433	27.3(26)	29.9 (2407)	0.5	0.5
Diagnostic cath	3321	2.1 (78)	3.8 (3243)	31.2	<0.0001	1272	7.0 (26)	10.6 (1246)	5.2	0.02
PCI	6253	2.3 (156)	2.9 (6,097)	9.5	0.002	4027	3.8 (100)	3.9 (3,927)	0.14	0.7
CABG	1857	2.8 (80)	4.1 (1,777)	11.1	0.0009	595	3.7 (19)	5.2 (576)	2.4	0.12

Output 4. Summary tables reporting Chi square and P values, generated by PROC SURVEYFREQ

PROCEDURES- MULTIVARIATE ANALYSES

This section focuses on the multivariate analyses and discusses the PROC SURVEYLOGISTC used to perform the adjusted analyses and obtain the point estimates along with the p- values.

Surveylogistic regression describes the relationship between a categorical response variable and a set of predictor variables. A categorical response variable can be a binary variable, an ordinal variable or a nominal variable; in this case it's a binary variable. Each type of categorical variables requires different techniques to model its relationship with the predictor variables. In a sample survey data, for a binary response variable, such as a response to a yes-no question, surveylogistic regression model are a commonly used model.

The next two examples show the independent effect of morbid obesity on any one coronary revascularization procedure utilization and overall in-hospital mortality in cases presenting with STEMI.

PROC SURVEYLOGISTIC

The independent variables adjusted in models include age, gender, peripheral vascular disease, paralysis, other neurologic disorders, chronic pulmonary disease, diabetes mellitus, diabetes mellitus with chronic complications, hypothyroidism, renal failure, liver disease, peptic ulcer disease, acquired immune deficiency syndrome, lymphoma, metastatic cancer, solid tumor without metastasis, rheumatoid arthritis, coagulopathy, weight loss, fluid and electrolyte disorders, chronic blood-loss anemia, iron deficiency anemia, alcohol abuse, drug abuse, psychoses, and hypertension.

Effect of morbid obesity on coronary revascularization procedure utilization

The following code and SAS output provide us the adjusted analyses results to show the effect of presence of morbid obesity on getting any coronary procedure.

```
proc surveylogistic data =Obesity;
strata NIS_stratum; cluster hospid;
 class ❷  female   (ref= first) morbidobesity (ref=first) ❸ diabetes htn_c  aids
 alcohol ANEMDEF arth race1(ref=first) income(ref=first) hosp_location
 h_contrl(ref=first) hosp_teach bldloss chf chrnlung coag depress drug hypothy liver
 lymph lytes mets neuro morbidobesity para perivasc psych pulmcirc renlfail tumor
 ulcer valve wghtloss  cararrhythmia/param=ref ❹;
 model anyproc1(event= '1') ❶=  age female  diabetes htn_c  aids alcohol ANEMDEF arth
race1 income  hosp_location h_contrl hosp_teach  TOTAL_DISC bldloss chf chrnlung coag
depress drug  hypothy liver lymph lytes mets neuro morbidobesity para perivasc psych
pulmcirc  renlfail tumor ulcer valve wghtloss cararrhythmia;
 where stemi=1; ❺
 weight discwt;
title 'Logi Reg anyproc vs Morbid obesity in STEMI cases';
run;
```

This code produces the point estimates and p values that are shown in Output 5.

In the MODEL statement he **event = '1'** option ❶ is specified in the PROC SURVEYLOGISTIC statement, to indicate which event to model. In this case the 'anyproc1= 1/yes' is modeled as an event. The other way of specifying the event to model is to use the 'event' option in MODEL statement using the quotes in the option as event = '1'.

The note is shown in Output5, about which event is modeled as **Probability modeled is "anyproc1" =1** ❻

In SAS, the **CLASS** statement ❷ is useful in creating the dummy variables for a categorical variable on-the-fly. There are various coding schemes from which to choose. The default coding for all the categorical variables in proc surveylogistic is the effect coding. Here we use dummy coding by using the **"param = ref"** option ❹and by specifying the comparison group by using **ref =** option ❸ after the variable name. For instance, for' morbid obesity' variable, the reference category used is "first" which is coded as "0" indicating those cases with absence of morbid obesity.

In Output 5, the first box shows the number of Observations Read and Number of Observations Used - The Number of Observations Used may be less than the Number of Observations Read if there are missing values for any variables in the equation. By default, SAS does a listwise deletion of incomplete cases.

The second box shows the response Variable profile and number of response Levels .

Parameter – They are the predictor/explanatory variables in the model and the intercept.

Estimate The estimates help here to understand the statistical aspect of the regression equation. The surveylogistic regression models the log odds of a positive response (probability modeled is anyproc1=1) as a linear combination the predictor variables. This is written as

$$\log[p / (1-p)] = b0 + b1*female + b2*morbidobesity + b3 *DM,$$

where p is the probability that Anyproc1 is 1. For our model, we have,

$$\log[p / (1-p)] = -8.16 + 0.3107*female + 0.8112* morbidobesity + 0.2384*Dm.$$

We can interpret the parameter estimates as follows: for a one unit change in the predictor variable, the difference in log-odds for a positive outcome is expected to change by the respective coefficient, given the other variables in the model are held constant.

MorbidObesity - This is the estimate regression coefficient comparing cases with morbid obesity ('1') with those without morbid obesity ('0').For a one unit change in MorbidObesity (from moving from 0 to 1), given the other variables in the model are held constant, the difference in log-odds for "anyproc" is expected to increase by 0.8112 unit, given the other variables in the model are held constant.

Point –estimate- The interpretation for the point estimate of Morbidobesity will be that for a one unit increase (from moving from morbid obesity (1) to No-morbid obesity (0), the odds of getting a "anyproc1" increases by (2.25-1)*100% = 125%. For female, from going from male to female (0 to 1), the odds of getting a "anyproc1" decreases by (1-0.73)*100% = 27%. The explanation for the female predictor is slightly different than the one for morbidobesity because the point estimate (Odds ratio) is less than 1, and actually renders the female gender in AMI patients as protective to undergo any cauterization procedure.

95% Wald Confidence Limits - For a given predictor variable with a level of 95% confidence the CI's would include the "true" population odds ratio. If the CI includes one, we'd fail to reject the null hypothesis that a particular regression coefficient equals zero and the odds ratio equals one, given the other predictors are in the model.

Number of Observations Read	26581
Number of Observations Used	25936
Sum of Weights Read	134032.3
Sum of Weights Used	130903.9

Response Profile

Ordered Value	DIED	Total Frequency	Total Weight
1	1	24344	122839.18
2	0	1592	8064.75

Probability modeled is anyproc1='1' ❻

Analysis of Maximum Likelihood Estimates

Parameter		DF	Estimate	Standard Error	Wald Chi-Square	Pr > ChiSq
Intercept		1	-8.1578	0.6639	151.0094	<.0001
AGE		1	-0.0533	0.00108	2435.9415	<.0001
FEMALE	1	1	-0.3107	0.0265	137.3793	<.0001
DM	0	1	0.2384	0.0289	67.9739	<.0001
Morbidobesity	1	1	0.8112	0.1095	54.9081	<.0001

Odds Ratio Estimates

Effect	Point Estimate	95% Wald Confidence Limits	
AGE	0.948	0.946	0.95
FEMALE 1 vs 0	0.733	0.696	0.772
DM 0 vs 1	1.269	1.199	1.343
morbidobesity 1 vs 0 ❼	2.251	1.41	3.59

Output 5. Portion Of The Output From PROC SURVEYLOGISTIC: Effect of morbid obesity on coronary revascularization procedure utilization

Output 5 displays part of the PROC SURVEYLOGISTIC output. It can be interpreted as below:

In those presenting with STEMIs, the likelihood of getting any procedure done was significantly higher in morbidly obese patients as compared to their counterparts. (OR 2.25 95 % CI 1.41-3.59) ❼. The CI does not include one and the p-value of <0.0001 confirms that we need to reject the null hypothesis and proves that 'morbid obesity' does have an independent predictive effect on coronary procedure utilization.

Effect of morbid obesity on in-hospital mortality

This code provides us the adjusted results to show the effect of presence of morbid obesity on in-hospital mortality.

```
proc surveylogistic data =Obesity;
strata NIS_stratum; cluster hospid;
 class   female  (ref= first) morbidobesity (ref=first) diabetes htn_c  aids alcohol
 ANEMDEF arth race1(ref=first) income(ref=first) hosp_location h_contrl(ref=first)
 hosp_teach bldloss chf chrnlung coag depress drug hypothy liver lymph lytes mets
 neuro morbidobesity para perivasc psych pulmcirc renlfail tumor ulcer valve wghtloss
 cararrhythmia/param=ref;
 model Died (event= '1')= age female  diabetes htn_c  aids alcohol ANEMDEF arth race1
income hosp_location h_contrl hosp_teach  TOTAL_DISC bldloss chf chrnlung coag depress
drug hypothy liver lymph lytes mets neuro morbidobesity para perivasc psych pulmcirc
 renlfail tumor ulcer valve wghtloss cararrhythmia;
weight discwt;
title 'Logi Reg Mortality vs Morbid obesity in AMI cases';
run;
```

Number of Observations Read	82173
Number of Observations Used	80475
Sum of Weights Read	413673.1
Sum of Weights Used	405440.5

Response Profile

Ordered Value	DIED	Total Frequency	Total Weight
1	1	4344	21924.3
2	0	76131	383516

Probability modeled is DIED='1'❽.

Analysis of Maximum Likelihood Estimates

Parameter		DF	Estimate	Standard Error	Wald Chi-Square	Pr > ChiSq
Intercept		1	-11.3651	43.509	0.0682	0.7939
AGE		1	0.0407	0.000638	4065.5311	<.0001
FEMALE	1	1	0.0031	0.0154	0.0408	0.8399
DM	0	1	-0.0753	0.0166	20.5462	<.0001
Morbidobesity	1	1	-0.1567	0.0467	11.2568	0.0008

Odds Ratio Estimates

Effect	Point Estimate	95% Wald Confidence Limits	
AGE	1.041	1.04	1.043
FEMALE 1 vs 0	1.003	0.973	1.034
DM 0 vs 1	0.927	0.898	0.958
morbidobesity 1 vs 0 ❾	0.855	0.70	0.937

Output 6. Portion Of The Output From PROC SURVEYLOGISTIC: Effect of morbid obesity on in-hospital mortality

In this PROC SURVEYLOGISTIC, the event is modeled as **Probability modeled is "died" =1/yes** ❽

For the overall sample presenting with AMI, the in-hospital mortality was 13 % lower for morbidly obese patients as compared to those who were not morbidly obese (OR 0.85, 95% CI 0.70-0.937) ❾. Similar results were observed in STEMI and NSTEMI cases.

In the Output 7, the summary tables show the compiled results obtained by PROC SURVEYLOGISTIC.
The section ❶ in Output 7 shows the odds ratio, confidence intervals and P values, obtained by modeling each invasive coronary procedure in a separate PROC SURVEYLOGISTI C. The section ❷ provides the same parameters obtained by modeling positive response for mortality post each coronary revascularization procedure in morbidly obese patients compared to their counterparts.

Thus the adjusted analyses using PROC SURVEYLOGISTIC shows that the morbidly obese cases are significantly more likely to undergo invasive coronary procedures (except PCI) compared to the non-morbid cases as shown in section ❶ . The section ❷ actually validates the Obesity paradox by displaying significantly lower likelihood of mortality in morbidly obese cases compared to their counterparts.

	AMI patients [a]		STEMI patients [b]		NSTEMI patients [c]	
❶Outcome Coronary. PROCEDURE UTILIZATION	Morbid obesity [j]		Morbid obesity		Morbid obesity	
Model used: Age, Gender, Race, Income, Hospital factors and 30 Elixhauser morbidities						
	Odds Ratio (95 % CI)	P - Value	Odds Ratio (95 % CI)	P - Value	Odds Ratio (95 % CI)	P - Value
Diagnostic cath	1.09 (1.04 -1.13)	<0.0001	1.12 (1.0003 - 1.25)	<0.0001	1.02 (0.98 - 1.07)	0.27
PCI	0.74 (0.72 - 0.77)	<0.0001	0.86 (0.80 - 0.94)	0.0003	0.78 (0.0.75-0.0.81)	<0.0001
CABG	1.66 (1.59 - 1.74)	<0.0001	1.60 (1.45 - 1.78)	<0.0001	1.61(1.53- 1.69)	<0.0001
Any procedure	1.08 (1.02 - 1.14)	0.01	2.25 (1.41 ╥ 3.59)	<0.0001	1.04 (0.98 - 1.10)	0.21

	AMI patients [a]		STEMI patients [b]		NSTEMI patients [c]	
❷ Outcome MORATLITY POST EACH Coronary PROCEDURE	Morbid obesity [j]		Morbid obesity		Morbid obesity	
Model used: Age, Gender, Race, Income, Hospital factors and 30 Elixhauser morbidities						
	Odds Ratio (95 % CI)	P - Value	Odds Ratio (95 % CI)	P - Value	Odds Ratio (95 % CI)	P - Value
Overall	0.85 (0.70- 0.937)	0.0008	0.90 (0.77- 1.07)	0.24	0.87 (0.78-0.98)	0.017
No procedure	0.79 (0.68- 0.92)	0.002	1.14 (0.69- 1.88)	0.61	0.83 (0.71- 0.98)	0.026
Diagnostic cath	0.75 (0.60 - 0.95)	0.018	0.97 (0.64 - 1.49)	0.9	0.76 (0.57 -1.02)	0.063
PCI	0.59 (0.51-0.69)	<.0001	1.06 (0.84 - 1.32)	0.64	0.91 (0.69- 1.20)	0.49

Output 7. Summary tables reporting Odds ratio, CI and P values, generated by PROC SURVEYLOGISTIC

CONCLUSION

In any analyses, knowing the data is the first step. PROC SURVEYFREQ is a very useful tools that can be used in exploring the basic information in a sample survey data, providing the distributional analyses and reporting the descriptive statistics for the categorical variables. Upon using the chi-square option it also generates the p-values that can be reported as a part of the bivariate results.

PROC SURVEYLOGISTIC for sample survey data is a powerful test that describes the relationship between a categorical response variable and a set of predictor variables and it generates the point estimates with the confidence interval. Both these procedures are easy and effective in obtaining bivariate and multivariate analyses results as well as confirming the pre-existing Obesity paradox.

ACKNOWLEDGMENTS

This paper is based on the paper listed below in the recommended readings and which I was a part of. I would like to thank the authors Dhoot Jashdeep, Tariq Shamail, Alpesh Amin, Pranav Patel and Shaista Malik, for their significant contribution in the original paper.

I would also like to thank all the SAS experts from the SAS discussion forum and SAS user group meeting for their valuable comments and suggestions while developing the SAS code.

- Institute of Digital Research Education UCLA; http://www.ats.ucla.edu/stat/sas/seminars/sas_surveylogistic/logistic1.htm
- Dhoot J, Tariq S, Erande A, Alpesh Amin, Pranav Patel, Shaista Malik. Effect of morbid obesity on in-hospital mortality and coronary revascularization outcomes after acute myocardial infarction in the United States. Am J Cardiol. 2013

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Name: Ashwini S Erande
Enterprise: University Of California Irvine
Address: 101 The City Drive South
City, State ZIP: Orange, CA, 92868
Work Phone: 714-456-7025
Fax: 714-456-7604
E-mail: aerande@uci.edu